



EXAMPLES OF VALIDATION RULES

Summary

Review examples of validation rules for various types of apps that you can use and modify for your own purposes. Validation rules verify that the data a user enters in a record meets the standards you specify before the user can save the record.

Review examples of validation rules for various types of apps that you can use and modify for your own purposes. *Validation rules* verify that the data a user enters in a record meets the standards you specify before the user can save the record.

Use the following samples for validation rules in Salesforce and Salesforce AppExchange apps, including:

Sample Account Address Validation Rules

Rules to maintain valid account addresses.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions by Context](#).

Canadian Billing Postal Code

Field	Value
Description:	Validates that the account Billing Zip/Postal Code is in the correct format if Billing Country is Canada.
Formula:	<pre>AND (OR (BillingCountry = "CAN", BillingCountry = "CA", BillingCountry = "Canada"), NOT (REGEX (BillingPostalCode, "(?i) [ABCEGHJKLMNPRSTVXY] \d[A-Z]? \s? \d[A-Z] \d) ?")))</pre>
Error Message:	Canadian postal code must be in A9A 9A9 format.
Error Location:	Billing Zip/Postal Code

Billing Zip Code Is in Billing State

Field	Value
Description:	Validates that the account <code>Billing Zip/Postal Code</code> is valid by looking up the first five characters of the value in a custom object called <code>Zip_Code__c</code> that contains a record for every valid ZIP code in the US. If the ZIP code isn't found in the <code>Zip_Code__c</code> object, or the <code>Billing State</code> doesn't match the corresponding <code>State_Code__c</code> in the <code>Zip_Code__c</code> object, an error is displayed.
Formula:	<pre>VLOOKUP(\$ObjectType.Zip_Code__c.Fields.City__c , \$ObjectType.Zip_Code__c.Fields.Name , LEFT(BillingPostalCode,5)) <> BillingCity</pre>
Error Message:	Billing Zip Code doesn't exist in specified Billing State.
Error Location:	Billing Zip/Postal Code

US Billing Zip Code

Field	Value
Description:	Validates that the account <code>Billing Zip/Postal Code</code> is in 99999 or 99999-9999 format if <code>Billing Country</code> is USA or US.
Formula:	<pre>AND(OR(BillingCountry = "USA", BillingCountry = "US"), NOT(Regex(BillingPostalCode, "\\d{5}(-\\d{4})?")))</pre> This example uses the REGEX function; see Shipping Zip Code if you aren't familiar with regular expressions.
Error Message:	ZIP code must be in 99999 or 99999-9999 format.
Error Location:	Billing Zip/Postal Code

Shipping Zip Code

Field	Value
Description:	Validates that the account Shipping Zip/Postal Code is in 99999 or 99999-9999 format if Shipping Country is USA or blank.
Formula:	<pre> AND (OR (ShippingCountry = "USA", ISBLANK (ShippingCountry)), OR (AND (LEN (ShippingPostalCode) <> 5, LEN (ShippingPostalCode) <> 10), NOT (CONTAINS ("0123456789", LEFT (ShippingPostalCode, 1))), NOT (CONTAINS ("0123456789", MID (ShippingPostalCode , 2, 1))), NOT (CONTAINS ("0123456789", MID (ShippingPostalCode , 3, 1))), NOT (CONTAINS ("0123456789", MID (ShippingPostalCode , 4, 1))), NOT (CONTAINS ("0123456789", MID (ShippingPostalCode , 5, 1))), AND (LEN (ShippingPostalCode) = 10, OR (MID (ShippingPostalCode , 6, 1) <> "-", NOT (CONTAINS ("0123456789", MID (ShippingPostalCode , 7, 1))), NOT (CONTAINS ("0123456789", MID (ShippingPostalCode , 8, 1))), NOT (CONTAINS ("0123456789", MID (ShippingPostalCode , 9, 1))), NOT (CONTAINS ("0123456789", MID (ShippingPostalCode , 10, 1))))))) </pre> This example interprets a blank country as the US. To use this example with other countries, remove the clause that checks the length of the country field. Also, validation rule criteria are case-sensitive, so this rule is only enforced when the country is blank or "USA" in all capital letters. The rule isn't enforced when the country is "usa." You can also validate ZIP codes using a regular expression; for an example of a formula using a regular expression, see REGEX.
Error Message:	ZIP code must be in 99999 or 99999-9999 format.
Error Location:	Shipping Zip/Postal Code

Valid Billing State (US)

Field	Value
Description:	Validates that the account Billing State/Province is a valid two-character abbreviation if Billing Country is US, USA, or blank.
Formula:	<pre> AND (OR(BillingCountry = "US", BillingCountry="USA", ISBLANK(BillingCountry)), OR (LEN(BillingState) < 2, NOT (CONTAINS ("AL:AK:AZ:AR:CA:CO:CT:DE:DC:FL:GA:HI:ID:" & "IL:IN:IA:KS:KY:LA:ME:MD:MA:MI:MN:MS:MO:MT:NE:NV:NH:" & "NJ:NM:NY:NC:ND:OH:OK:OR:PA:RI:SC:SD:TN:TX:UT:VT:VA:" & "WA:WV:WI:WY", BillingState)))) </pre> This example interprets a blank country as the US. To use this example with other countries, remove the clause that checks the length of the country field. Also, validation rule criteria are case-sensitive, so this rule is only enforced when the country is blank or "USA" in all capital letters. The rule isn't enforced when the country is "usa."
Error Message:	A valid two-letter state code is required.
Error Location:	Billing State/Province

Valid Billing Province (Canada)

Field	Value
Description:	Validates that the account Billing State/Province is a valid two-character abbreviation if Billing Country is CA or CAN.
Formula:	<pre>AND (OR(BillingCountry = "CA", BillingCountry="CAN"), OR(LEN(BillingState) < 2, NOT (CONTAINS("AB:BC:MB:NB:NL:NT:NS:NU:ON:PC:QC:SK:YT", BillingState))))</pre>
Error Message:	A valid two-letter province code is required.
Error Location:	Billing State/Province

Valid Shipping State

Field	Value
Description:	Validates that the account Shipping State/Province is a valid two-character abbreviation if Shipping Country is US, USA, or blank.
Formula:	<pre>AND (OR(ShippingCountry = "US", ShippingCountry="USA", ISBLANK(ShippingCountry)), OR(LEN(ShippingState) < 2, NOT (CONTAINS("AL:AK:AZ:AR:CA:CO:CT:DE:DC:FL:GA:HI:ID:" & "IL:IN:IA:KS:KY:LA:ME:MD:MA:MI:MN:MS:MO:MT:NE:NV:NH:" & "NJ:NM:NY:NC:ND:OH:OK:OR:PA:RI:SC:SD:TN:TX:UT:VT:VA:" & "WA:WV:WI:WY", ShippingState))))</pre> This example interprets a blank country as the US. To use this example with other countries, remove the clause that checks the length of the country field. Also, validation rule criteria are case-sensitive, so this rule is only enforced when the country is blank or "USA" in all capital letters. The rule isn't enforced when the country is "usa."
Error Message:	A valid two-letter state abbreviation is required.
Error Location:	Shipping State/Province

Valid Shipping Province (Canada)

Field	Value
Description:	Validates that the account Shipping State/Province is a valid two-character abbreviation, if Billing Country is CA or CAN.
Formula:	<pre>AND (OR(ShippingCountry = "CA", ShippingCountry="CAN"), OR(LEN(ShippingState) < 2, NOT(CONTAINS("AB:BC:MB:NB:NL:NT:NS:NU:ON:PC:QC:SK:YT", ShippingState))))</pre>
Error Message:	A valid two-letter province abbreviation is required.
Error Location:	Shipping State/Province

Valid Billing Country

Field	Value
Description:	Validates that the account Billing Country is a valid ISO 3166 two-letter code.
Formula:	<pre> OR (LEN(BillingCountry) = 1, NOT (CONTAINS ("AF:AX:AL:DZ:AS:AD:AO:AI:AQ:AG:AR:AM:" & "AW:AU:AZ:BS:BH:BD:BB:BY:BE:BZ:BJ:BM:BT:BO:" & "BA:BW:BV:BR:IO:BN:BG:BF:BI:KH:CM:CA:CV:KY:" & "CF:TD:CL:CN:CX:CC:CO:KM:CG:CD:CK:CR:CI:HR:" & "CU:CY:CZ:DK:DJ:DM:DO:EC:EG:SV:GQ:ER:EE:ET:FK:" & "FO:FJ:FI:FR:GF:PF:TF:GA:GM:GE:DE:GH:GI:GR:GL:" & "GD:GP:GU:GT:GG:GN:GW:GY:HT:HM:VA:HN:HK:HU:" & "IS:IN:ID:IR:IQ:IE:IM:IL:IT:JM:JP:JE:JO:KZ:KE:KI:" & "KP:KR:KW:KG:LA:LV:LB:LS:LR:LY:LI:LT:LU:MO:MK:" & "MG:MW:MY:MV:ML:MT:MH:MQ:MR:MU:YT:MX:FM:MD:MC:" & "MC:MN:ME:MS:MA:MZ:MM:MA:NR:NP:NL:AN:NC:NZ:NI:" & "NE:NG:NU:NF:MP:NO:OM:PK:PW:PS:PA:PG:PY:PE:PH:" & "PN:PL:PT:PR:QA:RE:RO:RU:RW:SH:KN:LC:PM:VC:WS:" & "SM:ST:SA:SN:RS:SC:SL:SG:SK:SI:SB:SO:ZA:GS:ES:" & "LK:SD:SR:SJ:SZ:SE:CH:SY:TW:TJ:TZ:TH:TL:TG:TK:" & "TO:TT:TN:TR:TM:TC:TV:UG:UA:AE:GB:US:UM:UY:UZ:" & "VU:VE:VN:VG:VI:WF:EH:YE:ZM:ZW", BillingCountry))) </pre>
Error Message:	A valid two-letter country code is required.
Error Location:	Billing Country

Sample Account Validation Rules

Validation rule examples for numeric account numbers, account number length, and annual revenue range.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions by Context](#).

Account Number Is Numeric

Field	Value
Description:	Validates that the <code>Account Number</code> is numeric if not blank.
Formula:	<pre>OR (ISBLANK (AccountNumber) , NOT (ISNUMBER (AccountNumber)))</pre>
Error Message:	Account Number is not numeric.
Error Location:	<code>Account Number</code>

Account Number Length

Field	Value
Description:	Validates that the <code>Account Number</code> is exactly seven digits (if it is not blank). The number seven is simply illustrative. You can change this to any number you like.
Formula:	<pre>AND (ISBLANK (AccountNumber) , LEN (AccountNumber) <> 7)</pre>
Error Message:	Account Number must be seven digits.
Error Location:	<code>Account Number</code>

Annual Revenue Range

Field	Value
Description:	Validates that the account <code>Annual Revenue</code> is not negative and does not exceed \$100 billion. This limit is designed to catch typos.
Formula:	<pre>OR (AnnualRevenue < 0 , AnnualRevenue > 100000000000)</pre>
Error Message:	Annual Revenue cannot exceed 100 billion.
Error Location:	<code>Annual Revenue</code>

Sample Call Center Validation Rules

Validation rules for requiring a conditional description, preventing cases from being reset, restricting case status, and more.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions](#).

Conditionally Require Description When Case Reason is "Other"

Field	Value
Description:	Validates that a custom field called <code>Other Reason</code> contains a value if a case has a <code>Case Reason</code> of "Other."
Formula:	<pre>AND (ISPICKVAL(Reason, "Other"), ISBLANK(Other_Reason__c))</pre>
Error Message:	Description of Other Reason is required.
Error Location:	<code>Other Reason</code>

Prevent Open Cases from Being Reset to New

Field	Value
Description:	If a case is already open, prevents the <code>Status</code> from being changed back to "New."
Formula:	<pre>AND (ISCHANGED(Status), NOT(ISPICKVAL(PRIORVALUE(Status), "New")), ISPICKVAL(Status, "New"))</pre>
Error Message:	Open case Status cannot be reset to New.
Error Location:	<code>Status</code>

Restrict Status of Re-Opened Cases

Field	Value
Description:	Validates that the case <code>Status</code> is "Re-opened" when a closed case is opened again.
Formula:	<pre>AND (ISCHANGED(Status), OR (ISPICKVAL(PRIORVALUE(Status), "Closed"), ISPICKVAL(PRIORVALUE(Status), "Closed in SSP"), NOT(ISPICKVAL(Status, "Re-Opened"))))</pre>
Error Message:	Closed case can only be changed to "Re-opened."
Error Location:	Status

Prevent Case Milestone Completion After Cases Are Closed

Field	Value
Description:	Validates that a milestone's <code>Completion Date</code> can't occur after the case's <code>Status</code> is Closed.
Formula:	<pre>Case.IsClosed = true</pre>
Error Message:	You can't complete a milestone after a case is closed.
Error Location:	Top of Page

Prevent Case Milestone Completion Before Case Creation Dates

Field	Value
Description:	Validates that the milestone's <code>Completion Date</code> has occurred after the case's <code>Date/Time Opened</code> .
Formula:	<pre>CompletionDate >= Case.CreatedDate && CompletionDate <= Case.ClosedDate</pre>
Error Message:	The milestone <code>Completion Date</code> must occur after the date the case was created and before the case was closed.
Error Location:	Top of Page

Sample Experience Cloud Site Validation Rules

Validation rule examples for various use cases, like preventing offensive language in questions, replies, ideas, and idea comments.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions by Context](#).

Preventing Offensive Language in Questions

Field	Value
Description:	Prevents users from entering offensive language in the <code>Title</code> and <code>Description</code> fields when asking a question.
Formula:	<pre>OR (CONTAINS (Title, 'darn'), CONTAINS (Body, 'darn'))</pre>
Error Message:	Question title or description contains offensive language.

Preventing Offensive Language in Replies

Field	Value
Description:	Prevents users from entering offensive language when replying to a question.
Formula:	<pre>OR (CONTAINS (Body, 'darn'), CONTAINS (Body, 'dang'))</pre>
Error Message:	Reply contains offensive language.

Preventing Offensive Language in Ideas

Field	Value
Description:	Prevents users from entering offensive language in the <code>Title</code> and <code>Description</code> fields when posting an idea.
Formula:	<pre>OR (CONTAINS (Title, 'darn'), CONTAINS (Body, 'darn'))</pre>
Error Message:	Idea title or description contains offensive language.

Preventing Offensive Language in Idea Comments

Field	Value
Description:	Prevents users from entering offensive language when posting a comment.
Formula:	<pre>OR (CONTAINS (CommentBody , 'darn'), CONTAINS (CommentBody, 'dang'))</pre>
Error Message:	Comment contains offensive language.

Sample Contact Validation Rules

Validation rules for various use cases, like requiring fields for mailing address, mailing street, and ZIP code.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions](#).

Mailing Address Fields Are Required

Field	Value
Description:	Validates that the contact Mailing Street, Mailing City, and Mailing Country are provided.
Formula:	<pre>OR (ISBLANK(MailingStreet), ISBLANK(MailingCity), ISBLANK(MailingCountry))</pre>
Error Message:	Mailing Street, City, and Country are required.
Error Location:	Top of Page

Mailing Street Is Required

Field	Value
Description:	Validates that the contact Mailing Street is provided.
Formula:	<pre>ISBLANK(MailingStreet)</pre>
Error Message:	Mailing Street is required.
Error Location:	Mailing Street

Mailing Zip Code

Field	Value
Description:	Validates that the contact Mailing Zip/Postal Code is in 99999 or 99999-9999 format if Mailing Country is USA or blank.
Formula:	<pre> AND (OR(MailingCountry = "USA", ISBLANK(MailingCountry)), OR (AND(LEN(MailingPostalCode) <>5, LEN(MailingPostalCode) <> 10), NOT(CONTAINS("0123456789", LEFT(MailingPostalCode, 1))), NOT(CONTAINS("0123456789", MID(MailingPostalCode , 2, 1))), NOT(CONTAINS("0123456789", MID(MailingPostalCode , 3, 1))), NOT(CONTAINS("0123456789", MID(MailingPostalCode , 4, 1))), NOT(CONTAINS("0123456789", MID(MailingPostalCode , 5, 1))), AND(LEN(MailingPostalCode) = 10, OR (MID(MailingPostalCode , 6, 1) <> "-", NOT(CONTAINS("0123456789", MID(MailingPostalCode , 7, 1))), NOT(CONTAINS("0123456789", MID(MailingPostalCode , 8, 1))), NOT(CONTAINS("0123456789", MID(MailingPostalCode , 9, 1))), NOT(CONTAINS("0123456789", MID(MailingPostalCode , 10, 1))))))) </pre> This example interprets a blank country as US. To use this example with other countries, remove the clause that checks the length of the country field. Also, validation rule criteria are case sensitive, so this rule is only enforced when the country is blank or "USA" in all capital letters. The rule is not enforced when the country is "usa." You can also validate zip codes using a regular expression; for an example of a formula using a regular expression, see REGEX.
Error Message:	Zip code must be in 99999 or 99999-9999 format.
Error Location:	Mailing Zip/Postal Code

Phone Number Has International Format

Field	Value
Description:	Validates that the <code>Phone</code> number begins with a plus sign (+) for country code. Note that this validation rule conflicts with the ten-digit rule.
Formula:	<code>LEFT (Phone, 1) <> "+"</code>
Error Message:	Phone number must begin with + (country code).
Error Location:	<code>Phone</code>

US Phone Number Has Ten Digits

Field	Value
Description:	Validates that the <code>Phone</code> number is in (999) 999-9999 format. This works by using the REGEX function to check that the number has ten digits in the (999) 999-9999 format.
Formula:	<code>NOT (REGEX (Phone, "\d*?(\d\d*?) {10}"))</code>
Error Message:	US phone numbers should be in this format: (999) 999-9999.
Error Location:	<code>Phone</code>

Sample Cross Object Validation Rules

Examples for three validation rules on opportunity products.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions by Context](#).

Discounts Must Be Within Range

The examples below work together to help you manage discount amounts for products and require a custom percent field on opportunity products called `Line Discount`. The examples below also require you to use price books and customize the `Product Family` field to include the following values: *Software*, *Consulting*, and *Training*.

Here is a validation rule for software discounts.

Field	Value
Description:	Prevents users from saving software products with a discount over 10 percent.

Field	Value
Formula:	<pre>AND(Line_Discount__c > 0.10, ISPICKVAL(Product2.Family, "Software"))</pre>
Error Message:	The discount must be 10% or less for software products.
Error Location:	Line Discount

Here is a validation rule for consulting discounts.

Field	Value
Description:	Prevents users from saving consulting products with a discount over 15 percent.
Formula:	<pre>AND(Line_Discount__c > 0.15, ISPICKVAL(Product2.Family, "Consulting"))</pre>
Error Message:	The discount must be 15% or less for consulting products.
Error Location:	Line Discount

Here is a validation rule for training discounts.

Field	Value
Description:	Prevents users from saving training products with a discount over 20 percent.
Formula:	<pre>AND(Line_Discount__c > 0.20, ISPICKVAL(Product2.Family, "Training"))</pre>
Error Message:	The discount must be 20% or less for training products.
Error Location:	Line Discount

Prevent Changing Opportunity Products on Closed Opportunities

This example consists of two validation rules: one on opportunity products and another on opportunities.

Field	Value
Description:	Prevents users from editing opportunity products after an opportunity is closed. Create the following validation rule example on opportunity products.
Formula:	<pre>OR (ISPICKVAL (Opportunity.StageName, "Closed Won"), ISPICKVAL (Opportunity.StageName, "Closed Lost"))</pre>
Error Message:	Cannot change opportunity products for closed opportunities.
Error Location:	Top of Page

The following validation rule is on opportunities.

Field	Value
Description:	Prevents users from deleting opportunity products after an opportunity is closed. Create the following validation rule example on opportunities. It uses a custom roll-up summary field on opportunities that counts the number of opportunity products on an opportunity.
Formula:	<pre>AND (OR (ISPICKVAL (StageName, "Closed Won"), ISPICKVAL (StageName, "Closed Lost")), Number_of_Line_Items__c < PRIORVALUE (Number_of_Line_Items__c))</pre>
Error Message:	Cannot delete opportunity products for closed opportunities.
Error Location:	Top of Page

Prevent Saving a Case When Account Does Not Have Support

Field	Value
Description:	Prevents users from saving a case for an account that does not have support. This example assumes you have a custom checkbox field on accounts called <code>Allowed Support</code> that tracks if the account has support.

Field	Value
Formula:	<code>Account.Allowed_Support__c = FALSE</code>
Error Message:	Unable to create cases for this account because it is not signed up for support.
Error Location:	Top of Page

Prevent Saving a Case When Contact is No Longer with the Company

Field	Value
Description:	Prevents users from saving an open case associated with a contact that is no longer with the company. This example uses a custom checkbox field on contacts called No Longer With Company.
Formula:	<code>AND(Contact.Not_Longer_With_Company__c, NOT(IsClosed))</code>
Error Message:	Unable to save this case because the related contact is no longer with the company. To continue, choose another contact.
Error Location:	Contact Name

Sample Date Validation Rules

Examples for date validation rules. For example, how to validate that the value of a custom field is a weekday, a Saturday or Sunday, that a custom date field contains a date within the current month and year, and more.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions by Context](#).

Date Must Be a Weekday

Field	Value
Description:	Validates that the value of a custom date field is a weekday (not Saturday or Sunday).
Formula:	<pre>CASE (MOD(My_Date__c - DATE(1900, 1, 7), 7), 0, 0, 6, 0, 1) = 0</pre>
Error Message:	Date must be a weekday.
Error Location:	My Date

Date Must Be a Weekend Day

Field	Value
Description:	Validates that the value of a custom date field is a Saturday or Sunday.
Formula:	<pre>CASE(MOD(My_Date__c - DATE(1900, 1, 7), 7), 0, 1, 6, 1, 0) = 0</pre>
Error Message:	Date must be a weekend day.
Error Location:	My Date

Date Must Be in the Current Month

Field	Value
Description:	Validates that a custom date field contains a date within the current month and year.
Formula:	<pre>OR (YEAR(My_Date__c) <> YEAR (TODAY()), MONTH(My_Date__c) <> MONTH (TODAY()))</pre>
Error Message:	Date must be in the current month.
Error Location:	My Date

Date Must Be in the Current Year

Field	Value
Description:	Validates that a custom date field contains a date within the current year.
Formula:	<code>YEAR (My_Date__c) <> YEAR (TODAY ())</code>
Error Message:	Date must be in the current year.
Error Location:	My Date

Date Must Be the Last Day of the Month

Field	Value
Description:	Validates whether a custom field called <code>My Date</code> is the last day of the month. To do this, it determines the date of the first day of the next month and then subtracts 1 day. It includes special case logic for December.
Formula:	<pre>DAY (My_Date__c) <> IF (Month (My_Date__c)=12, 31, DAY (DATE (YEAR (My_Date__c), MONTH (My_Date__c)+1, 1) - 1))</pre>
Error Message:	Date must be the last day of the month.
Error Location:	My Date

Date Must Be Within One Year of Today

Field	Value
Description:	Validates whether a custom field called <code>Follow-Up Date</code> is within one year of today's date. This example assumes a 365 day year. (It does not handle leap years.)
Formula:	<code>Followup_Date__c - TODAY () > 365</code>
Error Message:	Follow-Up Date must be within one year of today.
Error Location:	Follow-Up Date

Day of Month Cannot Be Greater Than 15

Field	Value
Description:	Validates that a custom field called <code>Begin Date</code> contains a date in the first 15 days of the specified month.
Formula:	<code>DAY(Begin_Date__c) > 15</code>
Error Message:	Begin Date cannot be after the 15th day of month.
Error Location:	<code>Begin Date</code>

End Date Cannot Be Before Begin Date

Field	Value
Description:	Validates that a custom field called <code>End Date</code> does not come before another custom field called <code>Begin Date</code> .
Formula:	<code>Begin_Date__c > End_Date__c</code>
Error Message:	End Date cannot be before Begin Date.
Error Location:	<code>Begin Date</code>

Expiration Date Cannot Be Before Close Date

Field	Value
Description:	Validates that a custom field called <code>Expiration Date</code> does not come before <code>Close Date</code> .
Formula:	<code>Expiration_Date__c < CloseDate</code>
Error Message:	Expiration Date cannot be before Close Date.
Error Location:	<code>Expiration Date</code>

Sample Number Validation Rules

Examples for how to validate that users can't save a time card record with more than 40 hours in a work week, numbers can't be negative, and even or odd numbers.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions](#).

Time Cards Must Total 40 Hours

Field	Value
Description:	Ensures that users cannot save a time card record with more than 40 hours in a work week. This example requires five custom fields on your custom object, one for each day of work.
Formula:	<pre>Monday_Hours__c + Tuesday_Hours__c + Wednesday_Hours__c + Thursday_Hours__c + Friday_Hours__c > 40</pre>
Error Message:	Your total hours cannot exceed 40.
Error Location:	Top of Page

Number Cannot Be Negative

Field	Value
Description:	Validates that a custom field called <code>Hours Worked</code> is not a negative number.
Formula:	<pre>Hours_Worked__c < 0</pre>
Error Message:	Hours Worked cannot be less than zero.
Error Location:	Hours Worked

Number Must Be Even

Field	Value
Description:	Validates that a custom field called <code>Ark Passengers</code> is a non-negative even number.
Formula:	<pre>OR (Ark_Passengers__c < 0, MOD (Ark_Passengers__c, 2) <> 0)</pre>
Error Message:	Ark Passengers must be a positive even number.
Error Location:	Ark Passengers

Number Must Be Odd

Field	Value
Description:	Validates that a custom field called <code>Socks_Found</code> is a non-negative odd number.
Formula:	<pre>OR (Socks_Found__c < 0, MOD(Socks_Found__c, 2) = 0)</pre>
Error Message:	Socks Found must be an odd number.
Error Location:	<code>Socks_Found</code>

Number Must Be a Multiple of Five

Field	Value
Description:	Validates that a custom field called <code>Multiple_of_5</code> is a multiple of five.
Formula:	<pre>MOD(Multiple_of_5__c, 5) <> 0</pre>
Error Message:	Number must be a multiple of five.
Error Location:	<code>Multiple_of_5</code>

Number Must Be an Integer

Field	Value
Description:	Validates that a custom field called <code>My_Integer</code> is an integer.
Formula:	<pre>FLOOR(My_Integer__c) <> My_Integer__c</pre>
Error Message:	This field must be an integer.
Error Location:	<code>My_Integer</code>

Number Must Be Between -50 and 50

Field	Value
Description:	Validates that a custom field called <code>Volume</code> is between -50 and 50.
Formula:	<code>ABS (Volume__c) > 50</code>
Error Message:	Volume must be between -50 and 50.
Error Location:	<code>Volume</code>

Number Range Validation

Field	Value
Description:	Validates that the range between two custom fields, <code>Salary Min</code> and <code>Salary Max</code> , is no greater than \$20,000.
Formula:	<code>(Salary_Max__c - Salary_Min__c) > 20000</code>
Error Message:	Salary range must be within \$20,000. Adjust the Salary Max or Salary Min values.
Error Location:	<code>Salary Max</code>

Percentage Must Be Between Zero and 100

Field	Value
Description:	Validates that a custom field called <code>Mix Pct</code> is between 0 and 100%. Note that percent fields are expressed divided by 100 in formulas (100% is expressed as 1; 50% is expressed as 0.5).
Formula:	<code>OR (Mix_Pct__c > 1.0, Mix_Pct__c < 0.0)</code>
Error Message:	Mix Pct must be between 0 and 100%.
Error Location:	<code>Mix Pct</code>

Sample Opportunity Management Validation Rules

Examples for Examples for how to validate custom fields and other fields on opportunities.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions](#).

Conditionally-Required Field Based on Opportunity Stage

Field	Value
Description:	Validates that a custom field called <code>Delivery Date</code> is provided if an opportunity has advanced to the Closed Won or Negotiation/Review stage.
Formula:	<pre>AND (OR (ISPICKVAL(StageName, "Closed Won"), ISPICKVAL(StageName, "Negotiation/Review")), ISBLANK(Delivery_Date__c))</pre>
Error Message:	Delivery Date is required for this stage.
Error Location:	<code>Delivery Date</code>

Close Date Cannot Be Prior to Current Month

Field	Value
Description:	Validates that the <code>Close Date</code> of an opportunity is not within a month prior to the current month. Note the use of <code>ISNEW</code> and <code>ISCHANGED</code> in this formula to ensure the condition is only checked when the opportunity is being created or the <code>Close Date</code> field is modified subsequently.
Formula:	<pre>AND (OR (ISNEW(), ISCHANGED(CloseDate)), CloseDate < DATE(YEAR(TODAY()), MONTH(TODAY()), 1))</pre>
Error Message:	Close Date cannot be prior to current month.
Error Location:	<code>Close Date</code>

Close Date Must Be a Future Date

Field	Value
Description:	Ensures that users do not change the <code>Close Date</code> of an opportunity to a day in the past.
Formula:	<code>SampleDate < TODAY()</code>
Error Message:	Close Date cannot be a day in the past.
Error Location:	<code>Close Date</code>

Discounts on Opportunities

Field	Value
Description:	Validates that a custom discount percent field is between 0 and 40%.
Formula:	<code>OR(Discount_Rate__c < 0, Discount_Rate__c > 0.40)</code>
Error Message:	The Discount Rate must not exceed 40%.
Error Location:	<code>Discount Rate</code>

High-Value Opportunity Must Be Approved Before Closed

Field	Value
Description:	Opportunities with amounts greater than \$50,000 require that a custom checkbox field called <code>Approved</code> is checked in order to change the stage to <code>Closed Won</code> or <code>Closed Lost</code> . To automate this, set field-level security on the <code>Approved</code> checkbox so that it can only be checked via a custom approval process (Enterprise Edition, Unlimited Edition, or Performance Edition).
Formula:	<pre>AND (OR (ISPICKVAL (StageName, "Closed Won"), ISPICKVAL (StageName, "Closed Lost"), (Amount > 50000), NOT (ISPICKVAL (Approval_Status__c, "Approved"))))</pre>
Error Message:	All high-value opportunities must be approved for closure. Click the Request Close button.
Error Location:	<code>Top of Page</code>

Opportunity Amount Cannot Exceed \$10 Million

Field	Value
Description:	Validates that opportunity Amount is positive and no more than \$10 million. This limit is designed to catch typos.
Formula:	<pre>OR (Amount < 0, Amount > 10000000)</pre>
Error Message:	Amount cannot exceed \$10 million.
Error Location:	Amount

Opportunity Check for Products

Field	Value
Description:	Validates that an opportunity has at least one opportunity product before users can save a change to an opportunity.
Formula:	<pre>NOT (OR (ISNEW () , HasOpportunityLineItem))</pre>
Error Message:	You must add products to this opportunity before saving.
Error Location:	Top of Page

Opportunity Must Have Products if Beyond “Needs Analysis” Stage

Field	Value
Description:	Validates that an opportunity has opportunity products before the Stage can move beyond Needs Analysis.
Formula:	<pre>AND (CASE(StageName, "Value Proposition", 1, "Id. Decision Makers", 1, "Perception Analysis", 1, "Proposal/Price Quote", 1, "Negotiation/Review", 1, "Closed Won", 1, 0) = 1, NOT (HasOpportunityLineItem))</pre>
Error Message:	Opportunity products are required to advance beyond the Needs Analysis stage.
Error Location:	Top of Page

Opportunity Name Format

Field	Value
Description:	Validates that an opportunity contains a hyphen as a way of enforcing an “[Account] - [Amount]” opportunity naming convention.
Formula:	<pre>FIND(" - ", Name) = 0</pre>
Error Message:	Opportunity Name should use “[Account] - [Amount]” format.
Error Location:	Opportunity Name

Prevent Sales Reps from Moving Opportunity Stage Backwards

Field	Value
Description:	Prevent sales reps from changing opportunity Stage "backwards" to specific values, once they have accepted the opportunity via a custom approval process. The approval process sets the custom Accepted Flag checkbox to True.
Formula:	<pre>AND(Accepted_Flag__c, OR (ISPICKVAL(StageName, "Stage 1"), ISPICKVAL(StageName, "Stage 2")))</pre>
Error Message:	Invalid stage for accepted opportunity.
Error Location:	Stage

Probability Must Be 100% for Won Opportunities

Field	Value
Description:	Validates that the probability of a won opportunity is properly set to 100%. This is useful for data cleanliness and reporting purposes.
Formula:	<pre>AND (ISPICKVAL(StageName, "Closed Won"), Probability <> 1))</pre>
Error Message:	Probability must be 100% for won opportunities.
Error Location:	Probability

Probability Must Be Zero for Lost Opportunities

Field	Value
Description:	Validates that the probability of a lost opportunity is properly set to zero. This is useful for data cleanliness and reporting purposes.
Formula:	<pre>AND (ISPICKVAL(StageName, "Closed Lost"), Probability <> 0))</pre>
Error Message:	Probability must be 0% for lost opportunities.
Error Location:	Probability

Project Start Date

Field	Value
Description:	Validates that a field is conditionally required based on the values of other fields. Use this validation formula to ensure that users include a <code>Project Start Date</code> for an opportunity that is closed/won.
Formula:	<pre>AND (ISPICKVAL (StageName, "Closed Won"), ISNULL (Project_Start_Date__c))</pre>
Error Message:	Project start date is required for won opportunities.
Error Location:	<code>Project Start Date</code>

Sample Quote Validation Rules

An example on how to validate a quote.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions by Context](#).

Display Error if Quote Line Item Discount Exceeds 40%

Field	Value
Description:	Shows an error if a quote line item's discount exceeds 40%.
Formula:	<pre>Discount > .40</pre>
Error Message:	The discount on this quote line item cannot exceed 40%.
Error Location:	<code>Discount</code> on quote

Sample User, Role, and Profile Validation Rules

Examples on how to validate custom user, role, and profile fields.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions](#).

Discount Percent Does Not Exceed Role-Based Limit

Field	Value
Description:	Validates that a custom field on opportunities called <code>Discount Percent</code> does not exceed a maximum value that varies depending on the user's role. The default maximum is 15%.
Formula:	<pre>Discount_Percent__c > VLOOKUP (\$ObjectType.Role_Limits__c.Fields.Limit__c, \$ObjectType.Role_Limits__c.Fields.Name, \$UserRole.Name)</pre>
Error Message:	Discount (%) exceeds limit allowed for your role.
Error Location:	<code>Discount Percent</code>

Expense Amount Does Not Exceed User's Max Allowed Expense

Field	Value
Description:	Validates a custom field called <code>Expense Amount</code> against a custom user field called <code>Max Allowed Expense</code> .
Formula:	<pre>Expense_Amount__c > \$User.Max_Allowed_Expense__c</pre>
Error Message:	Amount cannot exceed your maximum allowed expense.
Error Location:	<code>Expense Amount</code>

Only Record Owner Can Change Field

Field	Value
Description:	Ensures that only the record owner can make changes to a custom field called <code>Personal Goal</code> .
Formula:	<pre>AND (ISCHANGED(Personal_Goal__c), Owner <> \$User.Id)</pre>
Error Message:	Only record owner can change Personal Goal.
Error Location:	<code>Personal Goal</code>

Only Record Owner or Administrator Can Change Field

Field	Value
Description:	Ensures that a user can make changes to a custom field called <code>Personal Goal</code> only if the user is the record owner or has a custom profile of "Custom: System Admin."
Formula:	<pre>AND (ISCHANGED(Personal_Goal__c), Owner <> \$User.Id, \$Profile.Name <> "Custom: System Admin")</pre> \$Profile merge fields are only available in Enterprise, Unlimited, Performance, and Developer Editions.
Error Message:	Only record owner or administrator can change Personal Goal.
Error Location:	<code>Personal Goal</code>

Opportunity Close Date Can Only Be Back-Dated by Administrator

Field	Value
Description:	Validates that the <code>Close Date</code> of an opportunity does not fall prior to the current month, except for users who have a custom profile called "Custom: System Admin."
Formula:	<pre>AND (OR (ISNEW(), ISCHANGED(CloseDate), CloseDate < DATE(YEAR(TODAY()), MONTH(TODAY()), 1), \$Profile.Name <> "Custom: System Admin"))</pre> \$Profile merge fields are only available in Enterprise, Unlimited, Performance, and Developer Editions.
Error Message:	Close Date cannot be prior to current month.
Error Location:	<code>Close Date</code>

Miscellaneous Sample Validation Rules

Examples for how to validate certain number formats for credit card numbers or drivers licences.

For more information on any of the formula functions used in these examples, see [Formula Operators and Functions by Context](#).

Allow Number to Be Increased but Not Decreased

Field	Value
Description:	Allows a custom field called <code>Commit Amount</code> to be increased but not decreased after initial creation. This rule uses the <code>PRIORVALUE()</code> function to compare the updated value of the field to its value prior to update.
Formula:	<pre>PRIORVALUE(Commit_Amount__c) > Commit_Amount__c</pre>
Error Message:	Commit Amount cannot be decreased.
Error Location:	<code>Commit Amount</code>

California Driver's License

Field	Value
Description:	Ensures that a custom field called <code>Drivers License</code> is in the correct A9999999 format when the <code>Mailing State</code> is "CA".
Formula:	<pre>AND(MailingState = "CA", NOT(Regex(Drivers_License__c, "[A-Z]\\d{7}?")))</pre>
Error Message:	Invalid California driver's license format.
Error Location:	<code>Drivers License</code>

Force Users to Check "I Accept Terms" to Enter Certain Values

Field	Value
Description:	Uses a checkbox labeled "I accept terms" to force the user to select a checkbox in order to enter a value called <code>Number of Days</code> that exceeds their Paid Time Off (PTO) balance available.
Formula:	<pre>AND(NOT(I_accept_terms__c), Number_of_Days__c > \$User.PTO_Balance__c)</pre>
Error Message:	Request will cause a negative PTO balance. You must accept Negative PTO Balance terms.
Error Location:	<code>I accept terms</code>

Prohibit Changes to a Field After It Has Been Saved

Field	Value
Description:	Prevents users from changing a custom field called <code>Guaranteed Rate</code> after it has been saved initially.
Formula:	<pre>AND (NOT (ISNEW()), ISCHANGED(Guaranteed_Rate__c))</pre>
Error Message:	Guaranteed Rate cannot be changed.
Error Location:	Guaranteed Rate

Social Security Number Format

Field	Value
Description:	Validates that a custom text field called <code>SSN</code> is formatted in 999-99-9999 number format (if it is not blank). The pattern specifies: • Three single digits (0-9): <code>\d{3}</code> • A dash • Two single digits (0-9): <code>\d{2}</code> • A dash • Four single digits (0-9): <code>\d{4}</code>
Formula:	<pre>NOT (OR (ISBLANK(Social_Security_Number__c), REGEX(Social_Security_Number__c , "[0-9]{3}-[0-9]{2}-[0-9]{4}")))</pre>
Error Message:	SSN must be in this format: 999-99-9999.
Error Location:	SSN

Valid Currency

Field	Value
Description:	Validates selected currency against an explicit subset of active currencies in your organization using the <code>Currency</code> picklist. Use this example if you only allow some of the active currencies in your organization to be applied to certain types of records.
Formula:	<pre>CASE (CurrencyIsoCode, "USD", 1, "EUR", 1, "GBP", 1, "JPY", 1, 0) = 0</pre>
Error Message:	Currency must be USD, EUR, GBP, or JPY.
Error Location:	Currency

Valid Credit Card Number

Field	Value
Description:	Validates that a custom text field called <code>Credit_Card_Number</code> is formatted in 9999-9999-9999-9999 or 9999999999999999 number format when it is not blank. The pattern specifies: • Four digits (0-9) followed by a dash: <code>\d{4}-</code> • The aforementioned pattern is repeated three times by wrapping it in <code>{3}</code> • Four digits (0-9) • The OR character (<code> </code>) allows an alternative pattern of 16 digits of zero through nine with no dashes: <code>\d{16}</code>
Formula:	<pre>NOT(REGEX(Credit_Card_Number__c , "(((\d{4}-){3}\d{4}) \d{16})?"))</pre>
Error Message:	Credit Card Number must be in this format: 9999-9999-9999-9999 or 9999999999999999.
Error Location:	Credit Card Number

Valid IP Address

Field	Value
Description:	Ensures that a custom field called <code>IP Address</code> is in the correct format, four 3-digit numbers (0-255) separated by periods.
Formula:	<pre>NOT (REGEX(IP_Address__c, "((25[0-5] 2[0-4][0-9] [01]?[0-9][0-9]?) \\.) {3} (25[0-5] 2[0-4][0-9] [01]?[0-9][0-9]?) \$"))</pre>
Error Message:	Error: IP Address must be in form 999.999.999.999 where each part is between 0 and 255.
Error Location:	IP Address

Website Extension

Field	Value
Description:	Validates a custom field called <code>Web Site</code> to ensure its last four characters are in an explicit set of valid website extensions.
Formula:	<pre>AND (RIGHT(Web_Site__c, 4) <> ".COM", RIGHT(Web_Site__c, 4) <> ".com", RIGHT(Web_Site__c, 4) <> ".ORG", RIGHT(Web_Site__c, 4) <> ".org", RIGHT(Web_Site__c, 4) <> ".NET", RIGHT(Web_Site__c, 4) <> ".net", RIGHT(Web_Site__c, 6) <> ".CO.UK", RIGHT(Web_Site__c, 6) <> ".co.uk")</pre>
Error Message:	Web Site must have an extension of .com, .org, .net, or .co.uk.
Error Location:	Web Site